USGS
science for a changing world

Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A2

DOCUMENTATION OF A COMPUTER PROGRAM TO SIMULATE AQUIFER-SYSTEM COMPACTION USING THE MODULAR FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By S.A. Leake and D.E. Prudic

This chapter supersedes U.S. Geological Survey Open-File Report 88-482

Book 6
Chapter A2

IBS1AL

Narrative for Module IBS1AL

This module allocates space in the X array (McDonald and Harbaugh, 1988, chap. 3, p. 22-23) for all arrays needed for the Interbed-Storage Package.

1. Print a message indentifying the package.
2. Check flag ISS to determine if simulation is steady-state or transient.
3. If simulation is steady-state, cancel the Interbed-Storage option and continue with the simulation.
4. Read flag for saving cell-by-cell storage changes (IIBSCB) and flag for output control of compaction, subsidence and preconsolidation head.
5. If option to save cell-by-cell storage terms is selected, print message giving unit number selected. If option to specify output control is is selected, print message.
6. Read one-dimensional array indicator (IBQ) to determine which model layers have interbed storage.
7. Print a message indicating which layers have interbed storage.
8. Allocate space for preconsolidation head array (HC), elastic storage capacity (SCE), inelastic storage capacity (SCV), and compaction (SUB). Space is allocated by computing the location of the first element of each array as the first unused element in the X array (ISUM). The locations for the arrays are stored in variables LCHC, LCSCE, LCSCV, and LCSUB.
9. Calculate and print the amount of space used by the Interbed-Storage Package. Calculate the total amount of space used in the X array. If the amount of space used exceeds the amount allocated (LENX), print a message and terminate simulation.
10. RETURN.

Flow chart for module IBS1AL

ISS is a steady-state flag.

If ISS ≠ 0, the simulation is steady-state.

If ISS = 0, the simulation is transient.

IIBSCB is a flag and a unit number.

If IIBSCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded.

If IIBSCB ≤ 0, cell-by-cell terms will not be recorded.

IIBSOC is a flag.

If IIBSOC > 0, output control flags for printing and recording subsidence, compaction, and preconsolidation head will be read each time step.

If IIBSOC ≤ 0, output control flags will not be read.

NAQL is the number of layers with interbed storage.

HC is the preconsolidation head array.

SCE is the elastic storage capacity array.

SCV is the inelastic storage capacity array.

SUB is the compaction array.



Program Listing for Module IBS1AL

```
      SUBROUTINE IBS1AL(ISUM,LENX,LCHC,LCSCE,LCSCV,LCSUB,
     1                  NCOL,NROW,NLAY,IIBSCB,IIBSOC,ISS,IN,IOUT)
C
C-----VERSION 1115 06JUN1988 IBS1AL
C     ******************************************************************
C     ALLOCATE ARRAY STORAGE FOR INTERBED STORAGE PACKAGE
C     ******************************************************************
C
C       SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION IBQ1(80)
      COMMON /IBSCOM/ IBQ(80)
C     ------------------------------------------------------------------
C
C1------IDENTIFY PACKAGE.
      WRITE(IOUT,1)IN
    1 FORMAT(1H0,'IBS1 -- INTERBED STORAGE PACKAGE, VERSION 1,',
     1     ' 06/02/88',' INPUT READ FROM UNIT',I3)
C
C2------CHECK TO SEE THAT INTERBED STORAGE OPTION IS APPROPRIATE
      IF(ISS.EQ.0) GO TO 100
C
C3------IF INAPPROPRIATE PRINT A MESSAGE & CANCEL OPTION.
      WRITE(IOUT,8)
    8 FORMAT(1X,'INTERBED STORAGE INAPPROPRIATE FOR STEADY-STATE',
     1 ' PROBLEM.',/,1X,'OPTION CANCELLED, SIMULATION CONTINUING.')
      IN=0
      RETURN
C
C4------READ FLAG FOR STORING CELL-BY-CELL STORAGE CHANGES AND
C4------FLAG FOR PRINTING AND STORING COMPACTION, SUBSIDENCE, AND
C4------CRITICAL HEAD ARRAYS.
  100 READ(IN,3) IIBSCB,IIBSOC
    3 FORMAT(2I10)
C
C5------IF CELL-BY-CELL TERMS TO BE SAVED THEN PRINT UNIT NUMBER.
      IF(IIBSCB.GT.0) WRITE(IOUT,105) IIBSCB
  105 FORMAT(1X,'CELL-BY-CELL FLOW TERMS WILL BE SAVED ON UNIT',I3)
C
C5A-----IF OUTPUT CONTROL FOR PRINTING ARRAYS IS SELECTED PRINT MESSAGE.
      IF(IIBSOC.GT.0) WRITE(IOUT,106)
  106 FORMAT(1X,'OUTPUT CONTROL RECORDS FOR IBS1 PACKAGE WILL BE ',
     1 'READ EACH TIME STEP.')
C
C6------READ INDICATOR AND FIND OUT HOW MANY LAYERS HAVE INTERBED STORAGE.
      READ(IN,110) (IBQ(K),K=1,NLAY)
  110 FORMAT(40I2)
      NAQL=0
      DO 120 K=1,NLAY
      IF(IBQ(K).LE.0) GO TO 120
      NAQL=NAQL+1
```

```
      IBQ1(NAQL)=K
  120 CONTINUE
C
C7------IDENTIFY WHICH LAYERS HAVE INTERBED STORAGE.
      WRITE(IOUT,130) (IBQ1(K),K=1,NAQL)
  130 FORMAT(1X,'INTERBED STORAGE IN LAYER(S) ',80I2)
C
C8------ALLOCATE SPACE FOR THE ARRAYS HC, SCE, SCV, AND SUB.
      IRK=ISUM
      NA=NROW*NCOL*NAQL
      LCHC=ISUM
      ISUM=ISUM+NA
      LCSCE=ISUM
      ISUM=ISUM+NA
      LCSCV=ISUM
      ISUM=ISUM+NA
      LCSUB=ISUM
      ISUM=ISUM+NA
C
C9------CALCULATE & PRINT AMOUNT OF SPACE USED BY PACKAGE.
  300 IRK=ISUM-IRK
      WRITE(IOUT,4)IRK
    4 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED FOR INTERBED STORAGE')
      ISUM1=ISUM-1
      WRITE(IOUT,5)ISUM1,LENX
    5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
      IF(ISUM1.GT.LENX)WRITE(IOUT,6)
    6 FORMAT(1X,'   ***X ARRAY MUST BE MADE LARGER***')
C
C10-----RETURN.
      RETURN
      END
```

List of Variables for Module IBS1AL

Variable	Range	Definition
IBQ	Package	DIMENSION (80) Layer flag for interbed storage: > 0, Layer has interbed storage. ≤ 0, Layer does not have interbed storage.
IBQ1	Module	DIMENSION (80) List of layer numbers of layers with interbed storage.
IIBSCB	Package	Flag and a unit number. > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see McDonald and Harbaugh, 1988, chap. 4, p. 14-15) is set. ≤ 0, cell-by-cell flow terms will not be recorded.

Variable	Range	Definition
IIBSOC	Package	Flag. > 0, output control will be read each time step for printing and recording subsidence, compaction and preconsolidation head. ≤ 0, subsidence will be printed at the end of each stress period.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IRK	Module	Before this module allocates space, IRK is set equal to ISUM. After allocation, IRK is subtracted from ISUM to get the amount of space in the X array allocated by this module.
ISS	Global	Flag. = 0, simulation is transient. ≠ 0, simulation is steady state.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1	Module	ISUM+1
K	Module	Index
LCHC	Package	Location in the X array of the first element of array HC.
LCSCE	Package	Location in the X array of the first element of array SCE.
LCSCV	Package	Location in the X array of the first element of array SCV.
LCSUB	Package	Location in the X array of the first element of array SUB.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program.
NA	Module	Number of cells in all layers with interbed storage.
NAQL	Module	Number of layers with interbed storage.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.

IBS1RP

Narrative for Module IBS1RP

This module reads storage properties and preconsolidation-head arrays that are used by the package.

1. Read preconsolidation head for all layers with interbed storage. Read elastic and inelastic storage coefficients for all layers with interbed storage. Storage properties initially read in are the product of specific storage and thickness. Read starting compaction for all layers with interbed storage.

2. Loop through all cells in the model grid with interbed storage.

3. Multiply storage coefficients by area of the cell to obtain storage capacity.

4. Make sure that preconsolidation head at each node is equal to or greater than the starting head. If preconsolidation head is less than the starting head, set it equal to the starting head value.

5. Initialize and read flags that specify when subsidence, compaction, and preconsolidation head will be printed and saved on disk. Print messages indicating which options were selected.

6. RETURN.

Flow chart for module IBS1RP

LOC is a pointer to elements in the Interbed-Storage Package arrays.

LOCH is a pointer to elements in the HNEW array.

HC is the preconsolidation head array.

HNEW is the array containing computed head for the end of the time step.



Program Listing for Module IBS1RP

```
      SUBROUTINE IBS1RP(DELR,DELC,HNEW,HC,SCE,SCV,SUB,NCOL,NROW,
     1                  NLAY,NODES,IIBSOC,ISUBFM,ICOMFM,IHCFM,
     2                  ISUBUN,ICOMUN,IHCUN,IN,IOUT)
C
C-----VERSION 1117 02JUN1988 IBS1RP
C     ******************************************************************
C     READ INTERBED STORAGE DATA
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 ANAME
      DOUBLE PRECISION HNEW
      DIMENSION HNEW(NODES),HC(NODES),SCE(NODES),
     1       SCV(NODES),SUB(NODES),ANAME(6,4),
     2       DELR(NCOL),DELC(NROW)
C
      COMMON /IBSCOM/ IBQ(80)
C
      DATA ANAME(1,1),ANAME(2,1),ANAME(3,1),ANAME(4,1),ANAME(5,1),
     1  ANAME(6,1) /'   P','RECO','NSOL','IDAT','ION ','HEAD'/
      DATA ANAME(1,2),ANAME(2,2),ANAME(3,2),ANAME(4,2),ANAME(5,2),
     1  ANAME(6,2) /'ELAS','TIC ','INTE','RBED',' STO','RAGE'/
      DATA ANAME(1,3),ANAME(2,3),ANAME(3,3),ANAME(4,3),ANAME(5,3),
     1  ANAME(6,3) /' VIR','GIN ','INTE','RBED',' STO','RAGE'/
      DATA ANAME(1,4),ANAME(2,4),ANAME(3,4),ANAME(4,4),ANAME(5,4),
     1  ANAME(6,4) /'    ',' STA','RTIN','G CO','MPAC','TION'/
C     ------------------------------------------------------------------
C
C1------READ IN STORAGE AND CRITICAL HEAD ARRAYS
      NIJ=NROW*NCOL
      KQ=0
      DO 60 K=1,NLAY
      IF(IBQ(K).LE.0) GO TO 60
      KQ=KQ+1
      LOC=1+(KQ-1)*NIJ
      CALL U2DREL(HC(LOC),ANAME(1,1),NROW,NCOL,K,IN,IOUT)
      CALL U2DREL(SCE(LOC),ANAME(1,2),NROW,NCOL,K,IN,IOUT)
      CALL U2DREL(SCV(LOC),ANAME(1,3),NROW,NCOL,K,IN,IOUT)
      CALL U2DREL(SUB(LOC),ANAME(1,4),NROW,NCOL,K,IN,IOUT)
   60 CONTINUE
C
C2------LOOP THROUGH ALL CELLS WITH INTERBED STORAGE.
      KQ=0
      DO 80 K=1,NLAY
      IF(IBQ(K).LE.0) GO TO 80
      KQ=KQ+1
      NQ=(KQ-1)*NIJ
      NK=(K-1)*NIJ
      DO 70 IR=1,NROW
      NQR=NQ+(IR-1)*NCOL
```

```
      NKR=NK+(IR-1)*NCOL
      DO 70 IC=1,NCOL
      LOC=NQR+IC
      LOCH=NKR+IC
C
C3------MULTIPLY STORAGE BY AREA TO GET STORAGE CAPACITY.
      AREA=DELR(IC)*DELC(IR)
      SCE(LOC)=SCE(LOC)*AREA
      SCV(LOC)=SCV(LOC)*AREA
C
C4------MAKE SURE THAT PRECONSOLIDATION HEAD VALUES
C4------ARE CONSISTANT WITH STARTING HEADS.
      IF(HC(LOC).GT.HNEW(LOCH)) HC(LOC)=HNEW(LOCH)
   70 CONTINUE
   80 CONTINUE
C
C5------INITIALIZE AND READ OUTPUT FLAGS.
      ICOMFM=0
      ISUBFM=0
      IHCFM=0
      ICOMUN=0
      ISUBUN=0
      IHCUN=0
      IF(IIBSOC.LE.0) GO TO 200
      READ(IN,100) ISUBFM,ICOMFM,IHCFM,ISUBUN,ICOMUN,IHCUN
  100 FORMAT(6I10)
      WRITE(IOUT,110) ISUBFM,ICOMFM,IHCFM
  110 FORMAT(1HO,'    SUBSIDENCE PRINT FORMAT IS NUMBER',I4/
     1          '    COMPACTION PRINT FORMAT IS NUMBER',I4/
     2          ' CRITICAL HEAD PRINT FORMAT IS NUMBER',I4)
      IF(ISUBUN.GT.0) WRITE(IOUT,120) ISUBUN
  120 FORMAT(1HO,'    UNIT FOR SAVING SUBSIDENCE IS',I4)
      IF(ICOMUN.GT.0) WRITE(IOUT,130) ICOMUN
  130 FORMAT(1H ,'    UNIT FOR SAVING COMPACTION IS',I4)
      IF(IHCUN.GT.0)  WRITE(IOUT,140) IHCUN
  140 FORMAT(1H ,' UNIT FOR SAVING CRITICAL HEAD IS',I4)
C
C6------RETURN
  200 RETURN
      END
```

List of Variables for Module IBS1RP

Variable	Range	Definition
ANAME	Module	Labels for printout of input arrays.
AREA	Module	Area of cell.
DELC	Global	DIMENSION (NROW), Cell dimensions in the column direction. DELC(I) contains the width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimensions in the row direction. DELC(J) contains the width of column J.
HC	Package	DIMENSION (NCOL,NROW,NAQL), Preconsolidation head in each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBQ	Package	DIMENSION (80) Layer flag for interbed storage: > 0, Layer has interbed storage. ≤ 0, Layer does not have interbed storage.
IC	Module	Index for columns.
ICOMFM	Package	Code for format in which compaction will be printed.
ICOMUN	Package	Unit number on which an unformatted record containing compaction should be recorded.
IHCFM	Package	Code for format in which preconsolidation head will be printed.
IHCUN	Package	Unit number on which an unformatted record containing preconsolidation head should be recorded.
IIBSOC	Package	Flag. > 0, output control will be read each time step for printing and recording subsidence, compaction and preconsolidation head. ≤ 0, subsidence will be printed at the end of each stress period.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
SUBFM	Package	Code for format in which subsidence will be printed.

Variable	Range	Definition
ISUBUN	Package	Unit number on which an unformatted record containing subsidence should be recorded.
K	Module	Index of all model layers.
KQ	Module	Index of model layers with interbed storage
LOC	Module	Pointer to parts of SUB, HC, SCE, and SCV arrays.
LOCH	Module	Pointer to parts of HNEW array.
NCOL	Global	Number of columns in the grid.
NIJ	Module	Number of cells in a layer.
NK	Module	Number of cells preceding a particular layer in the HNEW array.
NKR	Module	Number of cells preceding a particular row in the HNEW array.
NLAY	Global	Number of layers in the grid.
NODES	Global	Number of cells (nodes) in the finite-difference grid.
NQ	Module	Number of cells preceding a particular layer in SUB, HC, SCE, and SCV arrays.
NQR	Module	Number of cells preceding a particular row in SUB, HC, SCE, and SCV arrays.
NROW	Global	Number of rows in the grid.
SCE	Package	DIMENSION (NCOL,NROW,NAQL), Elastic storage capacity of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
SCV	Package	DIMENSION (NCOL,NROW,NAQL), Inelastic storage capacity of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
SUB	Package	DIMENSION (NCOL,NROW,NAQL), Computed compaction of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.

IBS1FM

Narrative for Module IBS1FM

This module adds terms representing interbed storage to the accumulators in which the terms RHS and HCOF are formulated.

1. Initialize counter for layers with interbed storage. Calculate reciprocal of time-step length to reduce number of divisions later on.

2. Find all layers with interbed storage using IBQ indicator. For layers with interbed storage, loop through cells, skipping those outside of the flow region.

3. Determine applicable storage capacities at the start and end of time step so that storage changes for cell may be apportioned between elastic and inelastic components.

4. Add contributions from interbed storage to the RHS and HCOF accumulators.

5. RETURN.

Flow chart for module IBS1FM

TLED is the reciprocal of the time-step length.

KQ is an index of model layers with interbed storage.

IBQ is a layer flag.
IBQ > 0 means that model layer has interbed storage.
IBQ ≤ 0 means that model layer does not have interbed storage.

RHS is an accumulator in which the right-hand side of the finite-difference equation is formulated.

HCOF is an accumulator in which the coefficient of head in the cell is formulated.



Program Listing for Module IBS1FM

```
      SUBROUTINE IBS1FM(RHS,HCOF,HNEW,HOLD,HC,SCE,SCV,
     1                  IBOUND,NCOL,NROW,NLAY,DELT)
C
C-----VERSION 1223 02JUN1988 IBS1FM
C     ******************************************************************
C        ADD INTERBED STORAGE TO RHS AND HCOF
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DOUBLE PRECISION HNEW
      DIMENSION RHS(NCOL,NROW,NLAY),HCOF(NCOL,NROW,NLAY),
     1          IBOUND(NCOL,NROW,NLAY),HNEW(NCOL,NROW,NLAY),
     2          HOLD(NCOL,NROW,NLAY),HC(NCOL,NROW,NLAY),
     3          SCE(NCOL,NROW,NLAY),SCV(NCOL,NROW,NLAY)
C
      COMMON /IBSCOM/ IBQ(80)
C     ------------------------------------------------------------------
C
C1------INITIALIZE
       TLED=1./DELT
      KQ=0
C
C2------FIND LAYERS WITH INTERBED STORAGE
      DO 110 K=1,NLAY
      IF(IBQ(K).EQ.0) GO TO 110
      KQ=KQ+1
      DO 100 I=1,NROW
      DO 100 J=1,NCOL
      IF(IBOUND(J,I,K).LE.0) GO TO 100
C
C3------DETERMINE STORAGE CAPACITIES FOR CELL AT START AND END OF STEP
      RHO1=SCE(J,I,KQ)*TLED
      RHO2=RHO1
      HCTMP=HC(J,I,KQ)
      IF(HNEW(J,I,K).LT.HCTMP) RHO2=SCV(J,I,KQ)*TLED
C
C4------ADD APPROPRIATE TERMS TO RHS AND HCOF
      RHS(J,I,K)=RHS(J,I,K)-HCTMP*(RHO2-RHO1)-RHO1*HOLD(J,I,K)
      HCOF(J,I,K)=HCOF(J,I,K)-RHO2
  100 CONTINUE
  110 CONTINUE
C
C5------RETURN
      RETURN
      END
```

List of Variables for Module IBS1FM

Variable	Range	Definition
DELT	Global	Length of the current time step.
HC	Package	DIMENSION (NCOL,NROW,NAQL), Preconsolidation head in each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
HCOF	Global	DIMENSION (NCOL,NROW,NLAY), Coefficient of head in cell (J,I,K) in the finite-difference equation.
HCTMP	Module	Temporary HC(J,I,K).
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HOLD	Global	DIMENSION (NCOL,NROW,NLAY), Head at the start of the current time step.
I	Module	Index for rows.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IBQ	Package	DIMENSION (80), Layer flag for interbed storage: > 0, Layer has interbed storage. ≤ 0, Layer does not have interbed storage.
J	Module	Index for columns.
K	Module	Index for layers.
KQ	Module	Index of model layers with interbed storage.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
RHO1	Module	Elastic storage capacity divided by length of current time step.
RHO2	Module	Storage capacity at end of time step (elastic or inelastic) divided by length of current time step.

Variable	Range	Definition
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of finite-difference equation. RHS is an accumulation of terms from several different packages.
SCE	Package	DIMENSION (NCOL,NROW,NAQL), Elastic storage capacity of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
SCV	Package	DIMENSION (NCOL,NROW,NAQL), Inelastic storage capacity of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
TLED	Module	Reciprocal of the length of the current time step.

IBS1BD

Narrative for Module IBS1BD

This module calculates rates and volumes of water derived from interbed storage and updates the preconsolidation-head array.

1. Initialize the cell-by-cell flow term flag (IBD), and rate accumulators STOIN and STOUT.
2. Test to determine whether or not cell-by-cell flow terms are to be saved. If not, skip item 3, following.
3. Set cell-by-cell flow term flag (IBD) and clear the array (BUFF) in which they will be accumulated.
4. Loop through all cells in grid with interbed storage to perform rate and volume calculations.
5. Begin calculations of flow rates to or from interbed storage. If cell is outside of active flow region, skip it. Set temporary variables equal to previous head in cell, current head in cell, and preconsolidation head in cell.
6. Determine storage capacities at beginning and end of time step.
7. Calculate volume change in interbed storage for time step.
8. Calculate compaction associated with storage change. Accumulate compaction in array SUB.
9. If cell-by-cell flow rates are to be saved, calculate flow rate and add to BUFF.
10. Determine whether flow rate is into or out of storage, add magnitude of rate to STOIN OR STOUT.
11. If cell-by-cell flow terms are to be saved, call module UBUDSV to write the buffer (BUFF) onto disk.
12. Calculate rates from volume storage changes and time-step length. Move rates, volumes, and budget labels into arrays for printing.
13. Increment the budget term counter (MSUM).
14. Update the preconsolidation-head array. If head at end of time step is less than preconsolidation head, set preconsolidation head equal to head in cell.
15. RETURN.

STOIN is an accumulator in which all flows into the aquifer from interbed storage are added.

STOUT is an accumulator in which all flows out of the aquifer to interbed storage are added.

BUFF is an array in which values are stored as they are being gathered for printing or recording.

IBQ is a layer flag.
IBQ > 0 means that model layer has interbed storage.
IBQ ≤ 0 means that model layer does not have interbed storage.

SUB is the compaction array.

VBVL is a table of budget entries calculated by component-of-flow packages for use in calculating the volumetric budget.

VBNM is a table of labels for budget terms.

MSUM is the counter for budget entries and labels in VBVL and VBNM.

HC is the preconsolidation head array.

HNEW is the array containing computed head for the end of the time step.

Flow chart for module IBS1BD



Program Listing for Module IBS1BD

```
      SUBROUTINE IBS1BD(IBOUND,HNEW,HOLD,HC,SCE,SCV,SUB,DELR,DELC,
     1      NCOL,NROW,NLAY,DELT,VBVL,VBNM,MSUM,KSTP,KPER,IIBSCB,
     2      ICBCFL,BUFF,IOUT)
C-----VERSION 1224 02JUN1988 IBS1BD
C     ******************************************************************
C     CALCULATE VOLUMETRIC BUDGET FOR INTERBED STORAGE
C     ******************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 TEXT,VBNM
      DOUBLE PRECISION HNEW
      DIMENSION IBOUND(NCOL,NROW,NLAY),HOLD(NCOL,NROW,NLAY),
     1          HNEW(NCOL,NROW,NLAY),HC(NCOL,NROW,NLAY),
     2          SCE(NCOL,NROW,NLAY),SCV(NCOL,NROW,NLAY),
     3          SUB(NCOL,NROW,NLAY),VBVL(4,20),VBNM(4,20),
     4          BUFF(NCOL,NROW,NLAY),DELR(NCOL),DELC(NROW)
      DIMENSION TEXT(4)
C
      COMMON /IBSCOM/ IBQ(80)
      DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /'INTE','RBED',' STO','RAGE'/
C     ------------------------------------------------------------------
C
C1------INITIALIZE CELL-BY-CELL FLOW TERM FLAG (IBD) AND
C1------ACCUMULATORS (STOIN AND STOUT).
      IBD=0
      STOIN=0.
      STOUT=0.
C
C2------TEST TO SEE IF CELL-BY-CELL FLOW TERMS ARE NEEDED.
      IF(ICBCFL.EQ.0  .OR. IIBSCB.LE.0 ) GO TO 10
C
C3------CELL-BY-CELL FLOW TERMS ARE NEEDED SET IBD AND CLEAR BUFFER.
      IBD=1
      DO 5 IL=1,NLAY
      DO 5 IR=1,NROW
      DO 5 IC=1,NCOL
      BUFF(IC,IR,IL)=0.
    5 CONTINUE
C
C4------RUN THROUGH EVERY CELL IN THE GRID WITH INTERBED STORAGE.
   10 KQ=0
      TLED=1./DELT
      DO 110 K=1,NLAY
      IF(IBQ(K).EQ.0) GO TO 110
      KQ=KQ+1
      DO 100 I=1,NROW
      DO 100 J=1,NCOL
C
C5------CALCULATE FLOW FROM STORAGE (VARIABLE HEAD CELLS ONLY)
      IF(IBOUND(J,I,K).LE.0) GO TO 100
```

```
      HHOLD=HOLD(J,I,K)
      HHNEW=HNEW(J,I,K)
      HHC=HC(J,I,KQ)
C
C6------GET STORAGE CAPACITIES AT BEGINNING AND END OF TIME STEP.
      SBGN=SCE(J,I,KQ)
      SEND=SBGN
      IF(HHNEW.LT.HHC) SEND=SCV(J,I,KQ)
C
C7------CALCULATE VOLUME CHANGE IN INTERBED STORAGE FOR TIME STEP.
      STRG=HHC*(SEND-SBGN)+SBGN*HHOLD-SEND*HHNEW
C
C8------ACCUMULATE SUBSIDENCE ASSOCIATED WITH CHANGE IN STORAGE
      SUB(J,I,KQ)=SUB(J,I,KQ)+STRG/(DELR(J)*DELC(I))
C
C9------IF C-B-C FLOW TERMS ARE TO BE SAVED THEN ADD RATE TO BUFFER.
      IF(IBD.EQ.1) BUFF(J,I,K)=BUFF(J,I,K)+STRG*TLED
C
C10-----SEE IF FLOW IS INTO OR OUT OF STORAGE.
      IF(STRG)94,100,96
   94 STOUT=STOUT-STRG
      GO TO 100
   96 STOIN=STOIN+STRG
  100 CONTINUE
  110 CONTINUE
C
C11-----IF C-B-C FLOW TERMS WILL BE SAVED CALL UBUDSV TO RECORD THEM.
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IIBSCB,BUFF,NCOL,NROW,
     1                         NLAY,IOUT)
C
C12-----MOVE RATES,VOLUMES & LABELS INTO ARRAYS FOR PRINTING.
  200 VBVL(3,MSUM)=STOIN*TLED
      VBVL(4,MSUM)=STOUT*TLED
      VBVL(1,MSUM)=VBVL(1,MSUM)+STOIN
      VBVL(2,MSUM)=VBVL(2,MSUM)+STOUT
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
C13-----INCREMENT BUDGET TERM COUNTER
      MSUM=MSUM+1
C
C14-----UPDATE PRECONSOLIDATION HEAD ARRAY
      KQ=0
      DO 310 K=1,NLAY
      IF(IBQ(K).LE.0) GO TO 310
      KQ=KQ+1
      DO 300 I=1,NROW
      DO 300 J=1,NCOL
      IF(IBOUND(J,I,K).LE.0) GO TO 300
      HHNEW=HNEW(J,I,K)
      IF(HHNEW.LT.HC(J,I,KQ)) HC(J,I,KQ)=HHNEW
  300 CONTINUE
```

```
  310 CONTINUE
C
C15-----RETURN
      RETURN
      END
```

List of Variables for Module IBS1BD

Variable	Range	Definition
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
DELC	Global	DIMENSION (NROW), Cell dimensions in the column direction. DELC(I) contains the width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimensions in the row direction. DELC(J) contains the width of column J.
DELT	Global	Length of the current time step.
HC	Package	DIMENSION (NCOL,NROW,NAQL), Preconsolidation head in each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
HHC	Module	Temporary storage of HC(I,J,KQ).
HHNEW	Module	Temporary storage of HNEW(J,I,K) (single precision).
HHOLD	Module	Temporary storage of HOLD(J,I,K).
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HOLD	Global	DIMENSION (NCOL,NROW,NLAY), Head at the start of the current time step.
I	Module	Index for rows.
IBD	Module	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell

Variable	Range	Definition
IBQ	Package	DIMENSION (80), Layer flag for interbed storage: > 0, Layer has interbed storage. ≤ 0, Layer does not have interbed storage.
IC	Module	Index for columns.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be recorded for the current time step.
IIBSCB	Package	Flag and a unit number. > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see McDonald and Harbaugh, 1988, chap. 4, p. 14-15) is set. ≤ 0, cell-by-cell flow terms will not be recorded.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
J	Module	Index for columns.
K	Module	Index for layers.
KPER	Global	Stress period counter.
KQ	Module	Index of model layers with interbed storage.
KSTP	Global	Time step counter. Reset at the start of each stress period.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
SBGN	Module	Storage capacity for cell at beginning of time step.
SCE	Package	DIMENSION (NCOL,NROW,NAQL), Elastic storage capacity of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.

Variable	Range	Definition
SCV	Package	DIMENSION (NCOL,NROW,NAQL), Inelastic storage capacity of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
SEND	Module	Storage capacity for cell at end of time step.
STOIN	Module	Sum of decreases in interbed storage from individual cells.
STOUT	Module	Sum of increases in interbed storage from individual cells.
STRG	Module	Volume of flow into or out interbed storage for a single cell.
SUB	Package	DIMENSION (NCOL,NROW,NAQL), Computed compaction of each cell with interbed storage. NAQL is number of layers for which IBQ > 0.
TEXT	Module	Labels recorded along with cell-by-cell flow terms.
TLED	Module	Reciprocal of the length of the current time step.
VBNM	Global	DIMENSION(4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION(4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N), Rate for the current time step into the flow field. (2,N), Rate for the current time step out of the flow field. (3,N), Volume into the flow field during the simulation. (4,N), Volume out of the flow field during the simulation.